Apéndice
Álgebra lineal con wxMaxima

Objetivos

1. Definir matrices con wxMaxima.
2. Aplicar con wxMaxima operaciones con matrices.
3. Aplicar transformaciones elementales de matrices.
4. Calcular el determinante de una matriz cuadrada.
5. Interpretar las propiedades del determinante de una matriz cuadrada.
6. Plantear y resolver sistemas de ecuaciones lineales con wxMaxima.
7. Discutir y resolver sistemas de ecuaciones lineales.

Contenidos

A-1. Matrices. Operaciones con matrices.
A-2 Transformaciones elementales de matrices.
A-3 Rango de una matriz. Inversa de una matriz cuadrada regular. Factorización LU.
A-4. Determinante de una matriz cuadrada. Propiedades. Cálculo
A-3. Sistemas de ecuaciones lineales.

Referencias

AEM11 ALANINOS PRATS, J; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P. (2011)
Cálculo con wxMaxima.

G07 GLASNER, MOSES (2007)
A Maxima Guide for Calculus Students

RR05 REDONDO NEBLE, M. VICTORIA; RODRÍGUEZ GALVÁN, J. RAFAEL (2005)
Introducción a Maxima

RG07 RODRÍGUEZ GALVÁN, J. RAFAEL (2007)
Maxima con wxMaxima: software libre en el aula de matemáticas

RR08 RODRÍGUEZ RIOTORTO, MARIO (2008)
Primeros pasos en Maxima

RV09 RODRÍGUEZ, MARIO; VILLATE, JAIME (2009)
Manual de Maxima.

VR09 VALLEJO RODRÍGUEZ, JOSÉ ANTONIO (2009)
Cálculo diferencial con Maxima

A-1.- Matrices. Operaciones con matrices

Los contenidos de este apartado se desarrollan en el archivo **Apendice-1.wxm**.

En este apartado se verá la definición de matrices con wxMaxima y las diferentes operaciones con matrices.

A.1.1. Definición de matrices submatrices.

Definición de matrices:

```
(%i1) 'A=A:matrix(  [a11, a12, a13, a14],
                    [a21, a22, a23, a24],
                    [a31, a32, a33, a34]
                 );
```

(%o1) $A=\begin{bmatrix} a11 & a12 & a13 & a14 \\ a21 & a22 & a23 & a24 \\ a31 & a32 & a33 & a34 \end{bmatrix}$

Explicitar una fila o una columna de una matriz:

```
(%i2) a1=row(A,1);        a3=row(A,3);
```

(%o2) $a1=\begin{bmatrix} a11 & a12 & a13 & a14 \end{bmatrix}$

(%o3) $a3=\begin{bmatrix} a31 & a32 & a33 & a34 \end{bmatrix}$

```
(%i4) a[1]=col(A,1);       a[4]=col(A,4);
```

(%o4) $a_1=\begin{bmatrix} a11 \\ a21 \\ a31 \end{bmatrix}$

(%o5) $a_4=\begin{bmatrix} a14 \\ a24 \\ a34 \end{bmatrix}$

Obtención de submatrices por eliminación de una fila:

```
(%i6) 'A1=submatrix(1,A);
```

(%o6) $A1=\begin{bmatrix} a21 & a22 & a23 & a24 \\ a31 & a32 & a33 & a34 \end{bmatrix}$

Obtención de submatrices por eliminación de una columna:

```
(%i7) 'A2=submatrix(A,2);
```

$$(\%o7)\ A2=\begin{bmatrix} a11 & a13 & a14 \\ a21 & a23 & a24 \\ a31 & a33 & a34 \end{bmatrix}$$

Obtención de submatrices por eliminación de una fila y una columna:

```
(%i8) 'A12=submatrix(1,A,2);
```

$$(\%o8)\ A12=\begin{bmatrix} a21 & a23 & a24 \\ a31 & a33 & a34 \end{bmatrix}$$

Obtención de submatrices por eliminación de más de una fila y una columna:

```
(%i9) 'A123=submatrix(1, 2, A, 3);
```

$$(\%o9)\ A123=\begin{bmatrix} a31 & a32 & a34 \end{bmatrix}$$

Definición de un vector columna (obsérvese que previamente se ha de cargar el paquete "eigen"):

```
(%i10) load ("eigen")$
       X3=X3:columnvector([x1, x2, x3]);
```

$$(\%o11)\ X3=\begin{bmatrix} x1 \\ x2 \\ x3 \end{bmatrix}$$

Definición de una matriz por una función de los índices de fila y de columna:

```
(%i12) f[i,j]:=2*i+3*j;
       B1=B1:genmatrix(f,4,3);
```

$$(\%o12)\ f_{i,j}:=2\,i+3\,j$$

$$(\%o13)\ B1=\begin{bmatrix} 5 & 8 & 11 \\ 7 & 10 & 13 \\ 9 & 12 & 15 \\ 11 & 14 & 17 \end{bmatrix}$$

```
(%i14) g[i,j]:=1/(2*i+3*j-1);
       B2=B2:genmatrix(g,3,4);
```

(%o14) $g_{i,j} := \dfrac{1}{2\,i+3\,j-1}$

(%o15) $B2 = \begin{bmatrix} \frac{1}{4} & \frac{1}{7} & \frac{1}{10} & \frac{1}{13} \\ \frac{1}{6} & \frac{1}{9} & \frac{1}{12} & \frac{1}{15} \\ \frac{1}{8} & \frac{1}{11} & \frac{1}{14} & \frac{1}{17} \end{bmatrix}$

Definición de una matriz diagonal:

```
(%i16) 'D5=D5:diag_matrix(a11,a22,a33,a44,a55);
```

(%o16) $D5 = \begin{bmatrix} a11 & 0 & 0 & 0 & 0 \\ 0 & a22 & 0 & 0 & 0 \\ 0 & 0 & a33 & 0 & 0 \\ 0 & 0 & 0 & a44 & 0 \\ 0 & 0 & 0 & 0 & a55 \end{bmatrix}$

Definición de las matrices unidad (o identidad):

```
(%i17) I[2]=I2:ident(2);        I[4]=I4:ident(4);
```

(%o17) $I_2 = \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$

(%o18) $I_4 = \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$

Definición de una matriz nula de un cierto orden:

```
(%i19) O34=O34:zeromatrix(3,4);
```

(%o19) $O34 = \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$

Cálculo de la traza de una matriz cuadrada:

```
(%i20) 'C=C:matrix( [a11, a12], [a21, a22] );
       load("nchrpl")$ '(tr(C))=mattrace(C);
```

(%o20) $C=\begin{bmatrix} a11 & a12 \\ a21 & a22 \end{bmatrix}$

(%o22) $\mathrm{tr}(C)=a22+a11$

A.1.2 Operaciones algebraicas con matrices.

Suma de matrices de la misma estructura:

```
(%i1) 'A1=A1:matrix( [a11, a12, a13, a14], [a21, a22, a23, a24], [a31, a32, a33, a34] )$
      'A2=A2:matrix( [b11, b12, b13, b14], [b21, b22, b23, b24], [b31, b32, b33, b34] )$
(%i3) '(A1+A2)=A1+A2;
```

(%o3) $A2+A1=\begin{bmatrix} b11+a11 & b12+a12 & b13+a13 & b14+a14 \\ b21+a21 & b22+a22 & b23+a23 & b24+a24 \\ b31+a31 & b32+a32 & b33+a33 & b34+a34 \end{bmatrix}$

Producto de una matriz por un escalar:

```
(%i4) '(%lambda*A1)=%lambda*A1;
```

(%o4) $\lambda\, A1=\begin{bmatrix} \lambda\, a11 & \lambda\, a12 & \lambda\, a13 & \lambda\, a14 \\ \lambda\, a21 & \lambda\, a22 & \lambda\, a23 & \lambda\, a24 \\ \lambda\, a31 & \lambda\, a32 & \lambda\, a33 & \lambda\, a34 \end{bmatrix}$

Combinación lineal de matrices:

```
(%i5) '(%lambda*A1 + %mu*A2)=%lambda*A1 + %mu*A2;
```

(%o5) $\mu\, A2+\lambda\, A1=\begin{bmatrix} \mu\, b11+\lambda\, a11 & \mu\, b12+\lambda\, a12 & \mu\, b13+\lambda\, a13 & \mu\, b14+\lambda\, a14 \\ \mu\, b21+\lambda\, a21 & \mu\, b22+\lambda\, a22 & \mu\, b23+\lambda\, a23 & \mu\, b24+\lambda\, a24 \\ \mu\, b31+\lambda\, a31 & \mu\, b32+\lambda\, a32 & \mu\, b33+\lambda\, a33 & \mu\, b34+\lambda\, a34 \end{bmatrix}$

Para el producto de matrices hay que observar que en algunos programas el símbolo es el asterisco (*). Veamos que pasa con wxMaxima:

```
(%i6) '(A1*A2)=A1*A2;
```

(%o6) $A1\ A2=\begin{bmatrix} a11\, b11 & a12\, b12 & a13\, b13 & a14\, b14 \\ a21\, b21 & a22\, b22 & a23\, b23 & a24\, b24 \\ a31\, b31 & a32\, b32 & a33\, b33 & a34\, b34 \end{bmatrix}$

Obviamente esta operación no es el producto de matrices, para las que hace falta que el número de columnas de la primera sea igual al número de filas de la segunda. Veamos la sintaxis con wxMaxima:

```
(%i7) 'A3=A3:matrix( [c11, c12, c13], [c21, c22, c23], [c31, c32, c33], [c41, c42, c43] );
```

$$(\%o7)\ A3 = \begin{bmatrix} c11 & c12 & c13 \\ c21 & c22 & c23 \\ c31 & c32 & c33 \\ c41 & c42 & c43 \end{bmatrix}$$

```
(%i8) '(A1.A3)=A1.A3;
```

$$(\%o8)\ A1\ .\ A3 = \begin{bmatrix} a14\ c41 + a13\ c31 + a12\ c21 + a11\ c11 & a14\ c42 + a13\ c32 + a12\ c22 + a11\ c12 & a14\ c43 + a13\ c33 + a12\ c23 + a11\ c13 \\ a24\ c41 + a23\ c31 + a22\ c21 + a21\ c11 & a24\ c42 + a23\ c32 + a22\ c22 + a21\ c12 & a24\ c43 + a23\ c33 + a22\ c23 + a21\ c13 \\ a34\ c41 + a33\ c31 + a32\ c21 + a31\ c11 & a34\ c42 + a33\ c32 + a32\ c22 + a31\ c12 & a34\ c43 + a33\ c33 + a32\ c23 + a31\ c13 \end{bmatrix}$$

```
(%i9) '(A3.A1)=A3.A1;
```

$$(\%o9)\ A3\ .\ A1 = \begin{bmatrix} a31\ c13 + a21\ c12 + a11\ c11 & a32\ c13 + a22\ c12 + a12\ c11 & a33\ c13 + a23\ c12 + a13\ c11 & a34\ c13 + a24\ c12 + a14\ c11 \\ a31\ c23 + a21\ c22 + a11\ c21 & a32\ c23 + a22\ c22 + a12\ c21 & a33\ c23 + a23\ c22 + a13\ c21 & a34\ c23 + a24\ c22 + a14\ c21 \\ a31\ c33 + a21\ c32 + a11\ c31 & a32\ c33 + a22\ c32 + a12\ c31 & a33\ c33 + a23\ c32 + a13\ c31 & a34\ c33 + a24\ c32 + a14\ c31 \\ a31\ c43 + a21\ c42 + a11\ c41 & a32\ c43 + a22\ c42 + a12\ c41 & a33\ c43 + a23\ c42 + a13\ c41 & a34\ c43 + a24\ c42 + a14\ c41 \end{bmatrix}$$

Sirva esto para recordar que el producto de matrices no es conmutativo.

Producto de una matriz por un vector columna:

```
(%i10) load ("eigen")$
       X4=X4:columnvector([x1, x2, x3, x4])$
       A1X4=A1X4:A1.X4;
```

$$(\%o12)\ A1X4 = \begin{bmatrix} a14\ x4 + a13\ x3 + a12\ x2 + a11\ x1 \\ a24\ x4 + a23\ x3 + a22\ x2 + a21\ x1 \\ a34\ x4 + a33\ x3 + a32\ x2 + a31\ x1 \end{bmatrix}$$

Transposición de matrices:

```
(%i13) 'A1T=A1T:transpose(A1);
```

$$(\%o13)\ A1T = \begin{bmatrix} a11 & a21 & a31 \\ a12 & a22 & a32 \\ a13 & a23 & a33 \\ a14 & a24 & a34 \end{bmatrix}$$

A.1.3 Concatenación de matrices.

A partir de unas matrices iniciales se pueden construir nuevas matrices por concatenación de filas, es decir, escribir a continuación de cada fila de una matriz las filas de la segunda:

```
(%i14) 'A12=A12:addcol(A1, A2);
```

$$(\%o14)\ A12=\begin{bmatrix} a11 & a12 & a13 & a14 & b11 & b12 & b13 & b14 \\ a21 & a22 & a23 & a24 & b21 & b22 & b23 & b24 \\ a31 & a32 & a33 & a34 & b31 & b32 & b33 & b34 \end{bmatrix}$$

También se pueden construir nuevas matrices por concatenación de columnas, es decir, escribir a continuación de cada columna de una matriz las columnas de la segunda:

```
(%i15) 'A22=A22:addrow(A1, A2);
```

$$(\%o15)\ A22=\begin{bmatrix} a11 & a12 & a13 & a14 \\ a21 & a22 & a23 & a24 \\ a31 & a32 & a33 & a34 \\ b11 & b12 & b13 & b14 \\ b21 & b22 & b23 & b24 \\ b31 & b32 & b33 & b34 \end{bmatrix}$$

A.1.4 Matrices por bloques.

A partir de matrices cuadradas se pueden construir nuevas matrices con la estructura llamada “diagonal por bloques” o “bloque-diagonal”:

```
(%i16) load("diag")$
       A3=A3:matrix([a11, a12, a13], [a21, a22, a23], [a31, a32, a33])$
       A4=A4:matrix([b11, b12], [b21, b22])$
```

```
(%i19) diag([A3,A4]);
```

$$(\%o19)\ \begin{bmatrix} a11 & a12 & a13 & 0 & 0 \\ a21 & a22 & a23 & 0 & 0 \\ a31 & a32 & a33 & 0 & 0 \\ 0 & 0 & 0 & b11 & b12 \\ 0 & 0 & 0 & b21 & b22 \end{bmatrix}$$

(%i20) diag([A4,A3]);

$$(\%o20) \begin{bmatrix} b11 & b12 & 0 & 0 & 0 \\ b21 & b22 & 0 & 0 & 0 \\ 0 & 0 & a11 & a12 & a13 \\ 0 & 0 & a21 & a22 & a23 \\ 0 & 0 & a31 & a32 & a33 \end{bmatrix}$$

A-2.- Transformaciones elementales de matrices

Los contenidos de este apartado se desarrollan en el archivo **Apendice-2.wxm**.

A.2.1. Transformaciones elementales de filas y columnas

Se llaman transformaciones elementales de matrices las siguientes:

1. Intercambio de dos filas o dos columnas.
2. Multiplicación de una fila o columna por un escalar no nulo.
3. Sumar a una fila o columna un múltiple escalar (no nulo) de otra fila o columna.

Veamos la sintaxis con wxMaxima.

1. Intercambio de dos filas o dos columnas:

```
(%i1) 'A=A:matrix( [a11, a12, a13, a14],
                   [a21, a22, a23, a24],
                   [a31, a32, a33, a34]  );
```

$$(\%o1)\ A=\begin{bmatrix} a11 & a12 & a13 & a14 \\ a21 & a22 & a23 & a24 \\ a31 & a32 & a33 & a34 \end{bmatrix}$$

```
(%i2) A1=A1:rowswap(A, 1, 3);
```

$$(\%o2)\ A1=\begin{bmatrix} a31 & a32 & a33 & a34 \\ a21 & a22 & a23 & a24 \\ a11 & a12 & a13 & a14 \end{bmatrix}$$

Cabe decir que esta transformación se obtiene premultiplicando la matriz por la matriz unidad del mismo número de filas en la que se haya aplicado la transformación que se quiere aplicar en la matriz. En efecto:

```
(%i3) I3a=I3a:rowswap(ident(3), 1, 3);
```

$$(\%o3)\ I3a=\begin{bmatrix} 0 & 0 & 1 \\ 0 & 1 & 0 \\ 1 & 0 & 0 \end{bmatrix}$$

```
(%i4) I3a.A;
```

$$(\%o4)\ \begin{bmatrix} a31 & a32 & a33 & a34 \\ a21 & a22 & a23 & a24 \\ a11 & a12 & a13 & a14 \end{bmatrix}$$

Intercambio de columnas:

```
(%i5) A2=A2:columnswap(A, 1, 3);
```

$$(\%o5)\ A2 = \begin{bmatrix} a13 & a12 & a11 & a14 \\ a23 & a22 & a21 & a24 \\ a33 & a32 & a31 & a34 \end{bmatrix}$$

Esta transformación se obtiene postmultiplicando la matriz por la matriz unidad del mismo número de columnas en la que se haya aplicado la transformación que se quiere aplicar en la matriz. En efecto:

```
(%i6) I4a=I4a:columnswap(ident(4), 1, 3);
```

$$(\%o6)\ I4a = \begin{bmatrix} 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

```
(%i7) A.I4a;
```

$$(\%o7) \begin{bmatrix} a13 & a12 & a11 & a14 \\ a23 & a22 & a21 & a24 \\ a33 & a32 & a31 & a34 \end{bmatrix}$$

2. Multiplicación de una fila o columna por un escalar no nulo.

No hay una instrucción de wxMaxima que realice esta transformación. Para esto se ha de recurrir a la metodología de premultiplicación (filas) o postmultiplicación (columnas) comentada en el caso anterior:

```
(%i8) I3b=I3b:matrix( [1,0,0], [0, %lambda,0], [0,0,1] );
```

$$(\%o8)\ I3b = \begin{bmatrix} 1 & 0 & 0 \\ 0 & \lambda & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
(%i9) I3b.A;
```

$$(\%o9) \begin{bmatrix} a11 & a12 & a13 & a14 \\ \lambda\ a21 & \lambda\ a22 & \lambda\ a23 & \lambda\ a24 \\ a31 & a32 & a33 & a34 \end{bmatrix}$$

(%i10) I4b=I4b:matrix([1,0,0,0], [0,%lambda,0,0], [0,0,1,0], [0,0,0,1]);

$$(\%o10)\ I4b=\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & \lambda & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

(%i11) A.I4b;

$$(\%o11)\ \begin{bmatrix} a11 & \lambda\, a12 & a13 & a14 \\ a21 & \lambda\, a22 & a23 & a24 \\ a31 & \lambda\, a32 & a33 & a34 \end{bmatrix}$$

3. Sumar a una fila o columna un múltiple escalar (no nulo) de otra fila o columna.

(%i12) rowop(A,1,2,-%lambda);

$$(\%o12)\ \begin{bmatrix} \lambda\, a21+a11 & \lambda\, a22+a12 & \lambda\, a23+a13 & \lambda\, a24+a14 \\ a21 & a22 & a23 & a24 \\ a31 & a32 & a33 & a34 \end{bmatrix}$$

Veamos también su implementación con la premultiplicación por la transformación hecha en la matriz unidad:

(%i13) I3c=I3c:rowop(ident(3), 1, 2, -%lambda);

$$(\%o13)\ I3c=\begin{bmatrix} 1 & \lambda & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

(%i14) I3c.A;

$$(\%o14)\ \begin{bmatrix} \lambda\, a21+a11 & \lambda\, a22+a12 & \lambda\, a23+a13 & \lambda\, a24+a14 \\ a21 & a22 & a23 & a24 \\ a31 & a32 & a33 & a34 \end{bmatrix}$$

Y ahora con una columna:

(%i15) columnop(A,1,2,-%lambda);

$$(\%o15)\ \begin{bmatrix} \lambda\, a12+a11 & a12 & a13 & a14 \\ \lambda\, a22+a21 & a22 & a23 & a24 \\ \lambda\, a32+a31 & a32 & a33 & a34 \end{bmatrix}$$

```
(%i16) I4c=I4c:columnop(ident(4), 1, 2, -%lambda);
```

$$(\%o16)\ I4c=\begin{bmatrix}1 & 0 & 0 & 0\\ \lambda & 1 & 0 & 0\\ 0 & 0 & 1 & 0\\ 0 & 0 & 0 & 1\end{bmatrix}$$

```
(%i17) A.I4c;
```

$$(\%o17)\ \begin{bmatrix}\lambda\, a12+a11 & a12 & a13 & a14\\ \lambda\, a22+a21 & a22 & a23 & a24\\ \lambda\, a32+a31 & a32 & a33 & a34\end{bmatrix}$$

A.2.2. Obtención de matrices escalonadas y matrices escalonadas reducidas.

Estas matrices se obtienen con las instrucciones "triangularize", resultando una matriz triangular superior, y "echelon" que da una matriz triangular superior reducida. Veamos la sintaxis mediante unos ejemplos.

```
(%i1) 'A=A:matrix( [3, 4, -2, 6],
                   [5, 2, 0, -1],
                   [7, 2, 8, 9]  );
```

$$(\%o1)\ A=\begin{bmatrix}3 & 4 & -2 & 6\\ 5 & 2 & 0 & -1\\ 7 & 2 & 8 & 9\end{bmatrix}$$

```
(%i2) triangularize(A);
```

$$(\%o2)\ \begin{bmatrix}5 & 2 & 0 & -1\\ 0 & 14 & -10 & 33\\ 0 & 0 & 104 & 172\end{bmatrix}$$

```
(%i3) echelon(A);
```

$$(\%o3)\ \begin{bmatrix}1 & \frac{2}{5} & 0 & -\frac{1}{5}\\ 0 & 1 & -\frac{5}{7} & \frac{33}{14}\\ 0 & 0 & 1 & \frac{43}{26}\end{bmatrix}$$

```
(%i4) fpprintprec:5$
      echelon(A), numer;
```

(%o5) $\begin{bmatrix} 1 & 0.4 & 0 & -0.2 \\ 0 & 1 & -0.714 & 2.3571 \\ 0 & 0 & 1 & 1.6538 \end{bmatrix}$

A-3.- Rango de una matriz. Inversa de una matriz. Factorización LU.

Los contenidos de este apartado se desarrollan en el archivo **Apendice-3.wxm**.

A.3.1. Rango de una matriz

Para calcular el rango de una matriz hay que aplicar la instrucción “rank”:

```
(%i1) 'A=A:matrix( [3, 4, -2, 6],
                   [5, 2, 0, -1],
                   [7, 2, 8, 9]  );
```

$$(\%o1)\ A=\begin{bmatrix}3 & 4 & -2 & 6\\ 5 & 2 & 0 & -1\\ 7 & 2 & 8 & 9\end{bmatrix}$$

```
(%i2) r=rank(A);
```

$$(\%o2)\ r=3$$

A.3.2. Inversa de una matriz

Para calcular la inversa de una matriz cuadrada regular hay que verificar primero que su rango es el adecuado y aplicar después la instrucción “invert”:

```
(%i1) 'A=A:matrix([2,5],[1,3]);
```

$$(\%o1)\ A=\begin{bmatrix}2 & 5\\ 1 & 3\end{bmatrix}$$

```
(%i2) rank(A);
```

$$(\%o2)\ 2$$

```
(%i3) '(inv(A))=invA:invert(A);
```

$$(\%o3)\ inv(A)=\begin{bmatrix}3 & -5\\ -1 & 2\end{bmatrix}$$

Como se puede ver a continuación el cálculo mediante los métodos “manuales” tradicionales puede resultar complicado:

```
(%i4) fpprintprec:5$
      'C=C:matrix( [3.123, 4.543, -2.541, 4.543],
                   [5.027, 2.977, 0.891, -3.982],
                   [7.002, 2.804, 8.492, -2.304],
                   [2.804, 8.492, -2.304, 7.004]  );
```

$$(\%o5)\ C=\begin{bmatrix} 3.123 & 4.543 & -2.541 & 4.543 \\ 5.027 & 2.977 & 0.891 & -3.982 \\ 7.002 & 2.804 & 8.492 & -2.304 \\ 2.804 & 8.492 & -2.304 & 7.004 \end{bmatrix}$$

```
(%i6) invC=invert(C);
```

$$(\%o6)\ invC=\begin{bmatrix} 0.391 & 0.01 & 0.0535 & -0.23 \\ -0.34 & 0.115 & -0.0398 & 0.273 \\ -0.155 & -0.0932 & 0.103 & 0.0813 \\ 0.205 & -0.174 & 0.0609 & -0.0688 \end{bmatrix}$$

A.3.3. Factorización LU

Para calcular la descomposición LU de una matriz cuadrada regular, es decir, dos matrices la primera de les cuales (L) es triangular inferior y la segunda de las cuales (U) es triangular superior, que multiplicadas en este orden dan la matriz A, hay implementada en wxMaxima una instrucción que da el resultado correspondiente. Veamos un ejemplo.

```
(%i1) 'A=A:matrix([1,2,-3],[-3,-4,13],[2,1,-5]);
```

$$(\%o1)\ A=\begin{bmatrix} 1 & 2 & -3 \\ -3 & -4 & 13 \\ 2 & 1 & -5 \end{bmatrix}$$

Veamos que la matriz es regular:

```
(%i2) r=rank(A);
```

$(\%o2)\ r=3$

Y ahora apliquemos la instrucción que calcula estos factores:

```
(%i3) lu_factor(A, generalring)$
      get_lu_factors (%);
```

$$(\%o4)\ \left[\begin{bmatrix}1 & 0 & 0\\0 & 1 & 0\\0 & 0 & 1\end{bmatrix}, \begin{bmatrix}1 & 0 & 0\\-3 & 1 & 0\\2 & -\frac{3}{2} & 1\end{bmatrix}, \begin{bmatrix}1 & 2 & -3\\0 & 2 & 4\\0 & 0 & 7\end{bmatrix}\right]$$

La segunda matriz es la triangular inferior (L) y la tercera es la triangular superior (U). En efecto:

```
(%i5) L=L:matrix([1,0,0],[-3,1,0],[2,-3/2,1])$
      U=U:matrix([1,2,-3],[0,2,4],[0,0,7])$
      L.U;
```

$$(\%o7)\ \begin{bmatrix}1 & 2 & -3\\-3 & -4 & 13\\2 & 1 & -5\end{bmatrix}$$

A-4.- Determinante de una matriz cuadrada.

Los contenidos de este apartado se desarrollan en el archivo **Apendice-4.wxm**.

Veamos el desarrollo del cálculo del determinante de una matriz cuadrada de orden n, en los casos $n = 2$ y $n = 3$:

```
(%i1) 'A2=A2:matrix([a11,a12],[a21,a22]);    detA2=determinant(A2);
      det(A2)=determinant(A2);
```

(%o1) $A2=\begin{bmatrix} a11 & a12 \\ a21 & a22 \end{bmatrix}$

(%o2) $detA2=a11\,a22-a12\,a21$

(%o3) $det\left(\begin{bmatrix} a11 & a12 \\ a21 & a22 \end{bmatrix}\right)=a11\,a22-a12\,a21$

```
(%i4) 'A3=A3:matrix([a11,a12,a13],[a21,a22,a23],[a31,a32,a33]);    detA3=determinant(A3);    expand(%);
```

(%o4) $A3=\begin{bmatrix} a11 & a12 & a13 \\ a21 & a22 & a23 \\ a31 & a32 & a33 \end{bmatrix}$

(%o5) $detA3=a11\,(a22\,a33-a23\,a32)-a12\,(a21\,a33-a23\,a31)+a13\,(a21\,a32-a22\,a31)$

(%o6) $detA3=a11\,a22\,a33-a12\,a21\,a33-a11\,a23\,a32+a13\,a21\,a32+a12\,a23\,a31-a13\,a22\,a31$

Como es sabido, el determinante de una matriz triangular es igual al producto de los elementos de la diagonal. En efecto:

```
(%i7) 'B3=B3:matrix([a11,a12,a13],[0,a22,a23],[0,0,a33]);
      detB3=determinant(B3);
```

(%o7) $B3=\begin{bmatrix} a11 & a12 & a13 \\ 0 & a22 & a23 \\ 0 & 0 & a33 \end{bmatrix}$

(%o8) $detB3=a11\,a22\,a33$

```
(%i9) 'B4=B4:matrix([a11,a12,a13,a14],[0,a22,a23,a24],[0,0,a33,a34], [0,0,0,a44]  );
      detB4=determinant(B4);
```

(%o9) $B4=\begin{bmatrix} a11 & a12 & a13 & a14 \\ 0 & a22 & a23 & a24 \\ 0 & 0 & a33 & a34 \\ 0 & 0 & 0 & a44 \end{bmatrix}$

(%o10) $detB4=a11\,a22\,a33\,a44$

Veamos ahora un ejemplo numérico:

```
(%i1) 'C1=C1: matrix(   [1.01,2.02,0.99,0.01,0.01],
                        [1.02,1.01,-1.01,2.02,0.01],
                        [1.99,2.99,0.01,1.01,0.02],
                        [-1.01,0.02,1.02,-2.01,0.03],
                        [0.24, -0.972, 1.23, -1.954, 1.36]   );
```

$$(\%o1)\ C1=\begin{bmatrix}1.01 & 2.02 & 0.99 & 0.01 & 0.01\\ 1.02 & 1.01 & -1.01 & 2.02 & 0.01\\ 1.99 & 2.99 & 0.01 & 1.01 & 0.02\\ -1.01 & 0.02 & 1.02 & -2.01 & 0.03\\ 0.24 & -0.972 & 1.23 & -1.954 & 1.36\end{bmatrix}$$

```
(%i2) detC1=determinant(C1);
```

(%o2) detC1 = 2.795694905840001

Como es conocido, la inversa de una matriz cuadrada se puede calcular aplicando el concepto de determinante, mediante la matriz de adjuntos o cofactores transpuesta multiplicada por el inverso del determinante de la matriz. Así, por ejemplo:

```
(%i1) 'M1=M1: matrix(  [1,2,1,0],      [1,1,-1,2],      [2,3,0,1],      [-1,0,1,-2] );
```

$$(\%o1)\ M1=\begin{bmatrix}1 & 2 & 1 & 0\\ 1 & 1 & -1 & 2\\ 2 & 3 & 0 & 1\\ -1 & 0 & 1 & -2\end{bmatrix}$$

```
(%i2) detM1=detM1:determinant(M1);
```

(%o2) detM1 = 2

```
(%i3) adjM1=adjM1:adjoint(M1);
```

$$(\%o3)\ adjM1=\begin{bmatrix}-1 & -4 & 2 & -3\\ 0 & 2 & 0 & 2\\ 3 & 0 & -2 & -1\\ 2 & 2 & -2 & 0\end{bmatrix}$$

(%i4) inv_M1=invM1:(1/detM1)*adjM1;

$$(\%o4)\ inv\_M1=\begin{bmatrix} -\frac{1}{2} & -2 & 1 & -\frac{3}{2} \\ 0 & 1 & 0 & 1 \\ \frac{3}{2} & 0 & -1 & -\frac{1}{2} \\ 1 & 1 & -1 & 0 \end{bmatrix}$$

Esta matriz es la inversa de la matriz M1; en efecto:

(%i5) M1.invM1;

$$(\%o5)\ \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

A-5.- Sistemas de ecuaciones lineales.

Los contenidos de este apartado se desarrollan en el archivo **Apendice -5.wxm**.

A.5.1. Discusión y resolución de sistemas lineales.

La mecánica de resolución de sistemas de ecuaciones lineales sencillas con wxMaxima ya es conocida. Recordémosla con un ejemplo.

Ejemplo A.5.1. Resolver el sistema de ecuaciones lineales:

$$\begin{cases} 2x-4y=-2 \\ -3x+5y=1 \end{cases}$$

Introducimos las ecuaciones y asignemos a cada ecuación una referencia:

```
(%i1) Eq1:2*x-4*y=-2;
      Eq2:-3*x+5*y=1;
(%o1) 2 x-4 y=-2
(%o2) 5 y-3 x=1
```

Para resolver el sistema aplicamos la instrucción "linsolve" indicando las ecuaciones y las variables:

```
(%i3) linsolve([Eq1, Eq2], [x,y]);
(%o3) [x=3,y=2]
```

Interesa ahora ilustrar el procedimiento de discusión y resolución de sistemas de ecuaciones lineales aplicando el teorema de Rouché-Frobenius. Lo hacemos con dos ejemplos.

Ejemplo A.5.2. Discutir y resolver el sistema de ecuaciones lineales:

$$\begin{cases} x+2y+3z=1 \\ 2x-y+z=-1 \\ -x+3y-2z=2 \end{cases}$$

Como se ha dicho antes, hay que introducir las ecuaciones y asignar a cada ecuación una referencia:

```
(%i1) Eq1:  x + 2*y + 3*z = 1$
      Eq2:  2*x - y + z = -1$
      Eq3:  -x + 3*y - 2*z = 2$
```

Ahora calcularemos con wxMaxima las dos matrices asociadas al sistema lineal: la matriz del sistema formada por los coeficientes de las incógnitas o variables y a continuación la matriz ampliada, resultante de concatenar la matriz del sistema con la columna de los términos independientes. Matriz del sistema:

```
(%i4) A=A:coefmatrix( [Eq1, Eq2, Eq3] , [x,y,z]);
```

$$(\%o4)\ A = \begin{bmatrix} 1 & 2 & 3 \\ 2 & -1 & 1 \\ -1 & 3 & -2 \end{bmatrix}$$

Matriz ampliada: hay una instrucción de wxMaxima que da esta matriz, pero en una forma ligeramente diferente ya que lo hace pasando al primer miembro los términos independientes y, por lo tanto, cambiando su signo. En efecto:

```
(%i5) A0=augcoefmatrix( [Eq1, Eq2, Eq3] , [x,y,z]);
```

$$(\%o5)\ A0 = \begin{bmatrix} 1 & 2 & 3 & -1 \\ 2 & -1 & 1 & 1 \\ -1 & 3 & -2 & -2 \end{bmatrix}$$

Para calcular la matriz ampliada en la forma estándar, hay que definir el vector de términos independientes y hacer la concatenación con la matriz del sistema. Por lo tanto:

```
(%i6) load ("eigen")$     B=B:columnvector([1, -1, 2])$
      A0=A0:addcol( A, B) ;
```

$$(\%o8)\ A0 = \begin{bmatrix} 1 & 2 & 3 & 1 \\ 2 & -1 & 1 & -1 \\ -1 & 3 & -2 & 2 \end{bmatrix}$$

Ahora se ha de calcular el rango de cada una de estas matrices:

```
(%i9) r_A=rank(A);     r_A0=rank(A0);
```

$(\%o9)\ r\_A = 3$

$(\%o10)\ r\_A0 = 3$

Por lo tanto el sistema es compatible determinado. Para resolverlo por la metodología clásica se calcula la matriz escalonada reducida de la matriz ampliada:

```
(%i11) echelon(A0);
```

$$(\%o11)\ \begin{bmatrix} 1 & 2 & 3 & 1 \\ 0 & 1 & 1 & \frac{3}{5} \\ 0 & 0 & 1 & 0 \end{bmatrix}$$

De aquí se puede obtener la solución, así como con la instrucción de wxMaxima a estos efectos:

```
(%i12) linsolve( [Eq1, Eq2, Eq3] , [x,y,z] );
```

$$(\%o12)\ [x=-\frac{1}{5}, y=\frac{3}{5}, z=0]$$

Ejemplo A.5.3. Discutir y resolver el sistema de ecuaciones lineales:

$$\begin{cases} x+2y+3z=1 \\ 2x-y+z=-1 \end{cases}$$

Introducimos las ecuaciones y asignamos a cada ecuación una referencia:

```
(%i1) Eq1:  x + 2*y + 3*z = 1$
      Eq2:  2*x - y + z = -1$
```

Ahora calcularemos con wxMaxima las dos matrices asociadas al sistema lineal. Matriz del sistema:

```
(%i3) A=A:coefmatrix( [Eq1, Eq2] , [x,y,z]);
```

$$(\%o3)\ A=\begin{bmatrix} 1 & 2 & 3 \\ 2 & -1 & 1 \end{bmatrix}$$

Matriz ampliada:

```
(%i4) load ("eigen")$     B=B:columnvector([1, -1])$
      A0=A0:addcol( A, B) ;
```

$$(\%o6)\ A0=\begin{bmatrix} 1 & 2 & 3 & 1 \\ 2 & -1 & 1 & -1 \end{bmatrix}$$

Ahora calculemos el rango de cada una de estas matrices:

```
(%i7) r_A=rank(A);     r_A0=rank(A0);
(%o7) r_A = 2
(%o8) r_A0 = 2
```

Por lo tanto el sistema es compatible y simplemente indeterminado. Para resolverlo aplicamos la instrucción de wxMaxima:

```
(%i10) linsolve( [Eq1, Eq2] , [x,y,z] );
```

$$(\%o10)\ [x=-\frac{5\ \%r5+1}{5}, y=-\frac{5\ \%r5-3}{5}, z=\%r5]$$

Es decir la solución es:

$$\begin{cases} x=-\dfrac{5z+1}{5} \\ y=\dfrac{3-5z}{5} \\ z=z\in\mathbb{R} \quad \text{(parametro)} \end{cases}$$

Ejemplo A.5.4. Discutir y resolver el sistema de ecuaciones lineales:

$$\begin{cases} x+2y+3z=1 \\ 2x+4y+6z=-1 \end{cases}$$

Introducimos las ecuaciones y asignamos a cada ecuación una referencia:

```
(%i1) Eq1:  x + 2*y + 3*z = 1$
      Eq2:  2*x +4*y + 6*z = -1$
```

Ahora calcularemos con wxMaxima las dos matrices asociadas al sistema lineal. Matriz del sistema:

```
(%i3) A=A:coefmatrix( [Eq1, Eq2] , [x,y,z]);
```

$$(\%o3)\ A=\begin{bmatrix} 1 & 2 & 3 \\ 2 & 4 & 6 \end{bmatrix}$$

Matriz ampliada:

```
(%i4) load ("eigen")$     B=B:columnvector([1, -1])$
      A0=A0:addcol( A, B) ;
```

$$(\%o6)\ A0=\begin{bmatrix} 1 & 2 & 3 & 1 \\ 2 & 4 & 6 & -1 \end{bmatrix}$$

Ahora calculemos el rango de cada una de estas matrices:

```
(%i7) r_A=rank(A);     r_A0=rank(A0);
(%o7) r_A = 1
(%o8) r_A0 = 2
```

Por lo tanto el sistema es incompatible. Veamos qué sucede si se quiere resolver con wxMaxima:

```
(%i10) linsolve( [Eq1, Eq2] , [x,y,z] );
(%o10) []
```

A.5.2. Sistemas lineales con parámetros.

La discusión de un sistema de ecuaciones lineales a menudo involucra la existencia de uno o más parámetros en las ecuaciones; la tipología y la solución del sistema depende del valor de estos parámetros y hay que hacer la discusión correspondiente. Ilustraremos el procedimiento con un par de ejemplos.

Ejemplo A.5.5. Discutir y resolver en función del parámetro $a \in \mathbb{R}$ el sistema de ecuaciones lineales:

$$\begin{cases} 3x - 2y + z = 1 \\ 4x + y - 2z = 2 \\ 2x - 5y - az = 3 \end{cases}$$

Introducimos las ecuaciones y asignamos a cada ecuación una referencia:

```
(%i1) Eq1:  3*x - 2*y + z = 1$
      Eq2:  4*x + y  - 2* z = 2$
      Eq3:  2*x - 5*y - a*z  = 3$
```

Matriz del sistema y matriz ampliada:

```
(%i4) A=A:coefmatrix ( [Eq1, Eq2, Eq3] , [x,y,z] );
```

$$(\%o4)\ A = \begin{bmatrix} 3 & -2 & 1 \\ 4 & 1 & -2 \\ 2 & -5 & -a \end{bmatrix}$$

```
(%i5) load ("eigen")$     B=B:columnvector([1, 2, 3])$
      A0=A0:addcol( A, B) ;
```

$$(\%o7)\ A0=\begin{bmatrix}3 & -2 & 1 & 1\\ 4 & 1 & -2 & 2\\ 2 & -5 & -a & 3\end{bmatrix}$$

Triangularizamos la matriz ampliada:

```
(%i8) triangularize(A0);
```

$$(\%o8)\ \begin{bmatrix}3 & -2 & 1 & 1\\ 0 & 11 & -10 & 2\\ 0 & 0 & -11\,a-44 & 33\end{bmatrix}$$

Ahora tenemos dos situaciones:

1) Si $-11a-44=0$, es decir, si $a=-4$, entonces $rang(A)=2$; $rang(A_0)=3$ y por lo tanto el sistema es incompatible.
2) Si $-11a-44\neq 0$, es decir, si $a\neq -4$, entonces $rang(A)=rang(A_0)=3$ y por lo tanto el sistema es compatible determinado.

En efecto:

```
(%i9) A1=A1:matrix([3,-2,1],[4,1,-2],[2,-5,4]);   r_A1=rank(A1);
```

$$(\%o9)\ A1=\begin{bmatrix}3 & -2 & 1\\ 4 & 1 & -2\\ 2 & -5 & 4\end{bmatrix}$$

$$(\%o10)\ r\_A1=2$$

```
(%i11) A01=A01:matrix([3,-2,1,1],[4,1,-2, 2],[2,-5,4, 3]) ;      r_A01=rank(A01);
```

$$(\%o11)\ A01=\begin{bmatrix}3 & -2 & 1 & 1\\ 4 & 1 & -2 & 2\\ 2 & -5 & 4 & 3\end{bmatrix}$$

$$(\%o12)\ r\_A01=3$$

Solución del sistema en el segundo caso ($a\neq -4$):

```
(%i13) linsolve([Eq1, Eq2, Eq3], [x,y,z]);
```

$$(\%o13)\ [x=\frac{5\,a+11}{11\,a+44},y=\frac{2\,a-22}{11\,a+44},z=-\frac{3}{a+4}]$$

Ejemplo A.5.6. Discutir y resolver en función de los parámetros $a,b \in \mathbb{R}$ el sistema de ecuaciones lineales:

$$\begin{cases} ax+y+z=b \\ x+ay+z=b \\ x+y+az=b \end{cases}$$

Introducimos las ecuaciones y asignamos a cada ecuación una referencia:

```
(%i1) Eq1:  a*x + y + z = b$
      Eq2:  x + a*y  + z = b$
      Eq3:  x + y + a*z  = b$
```

Matriz del sistema y matriz ampliada:

```
(%i4) A=A:coefmatrix ( [Eq1, Eq2, Eq3] , [x,y,z] );
```

$$(\%o4)\ A=\begin{bmatrix} a & 1 & 1 \\ 1 & a & 1 \\ 1 & 1 & a \end{bmatrix}$$

```
(%i5) load ("eigen")$      B=B:columnvector([b, b, b])$
      A0=A0:addcol( A, B) ;
```

$$(\%o7)\ A0=\begin{bmatrix} a & 1 & 1 & b \\ 1 & a & 1 & b \\ 1 & 1 & a & b \end{bmatrix}$$

Discutiremos el rango de la matriz del sistema ya que sólo depende de un parámetro y lo haremos con su determinante, ya que es cuadrada:

```
(%i8) detA=determinant(A);
      expand(%);
```

$(\%o8)\ detA=a\,(a^2-1)-2\,a+2$

$(\%o9)\ detA=a^3-3\,a+2$

```
(%i10) solve(a^3-3*a+2=0, a);
```

$(\%o10)\ [a=-2, a=1]$

Ahora tenemos tres situaciones:

1) $a=1$;
2) $a=-2$;
3) $a\neq 1$ y $a\neq -2$.

<u>Primer caso</u>. La matriz del sistema y la matriz ampliada son:

```
(%i11) A1=A1:matrix([1,1,1],[1,1,1],[1,1,1]);
```

$$\text{(\%o11) } A1 = \begin{bmatrix} 1 & 1 & 1 \\ 1 & 1 & 1 \\ 1 & 1 & 1 \end{bmatrix}$$

```
(%i12) A01=A01:matrix([1,1,1,b],[1,1,1,b],[1,1,1,b]);
```

$$\text{(\%o12) } A01 = \begin{bmatrix} 1 & 1 & 1 & b \\ 1 & 1 & 1 & b \\ 1 & 1 & 1 & b \end{bmatrix}$$

Rango de estas matrices:

```
(%i13) r_A1=rank(A1);     r_A01=rank(A01);
(%o13) r_A1 = 1
(%o14) r_A01 = 1
```

Por lo tanto el sistema es compatible doblemente indeterminado. Solución del sistema:

```
(%i15) Eq11:  x + y + z = b$
       Eq21:  x + y + z = b$
       Eq31:  x + y + z = b$
(%i18) linsolve  (   [Eq11, Eq21, Eq31] , [x,y,z] );
solve: dependent equations eliminated: (2 3)
(%o18) [x = b - %r7 - %r6 , y = %r7 , z = %r6 ]
```

<u>Segundo caso</u>. La matriz del sistema y su rango son:

```
(%i19) A2=A2:matrix([-2,1,1],[1,-2,1],[1,1,-2]);     r_A2=rank(A2);
```

$$\text{(\%o19) } A2 = \begin{bmatrix} -2 & 1 & 1 \\ 1 & -2 & 1 \\ 1 & 1 & -2 \end{bmatrix}$$

```
(%o20) r_A2 = 2
```

Matriz ampliada:

```
(%i21) A02=A02:matrix([-2,1,1,b],[1,-2,1,b],[1,1,-2,b]);
```

$$\text{(\%o21) } A02 = \begin{bmatrix} -2 & 1 & 1 & b \\ 1 & -2 & 1 & b \\ 1 & 1 & -2 & b \end{bmatrix}$$

Triangularizamos la matriz:

```
(%i22) triangularize(A02);
```

$$(\%o22)\ \begin{bmatrix} -2 & 1 & 1 & b \\ 0 & 3 & -3 & -3\,b \\ 0 & 0 & 0 & 9\,b \end{bmatrix}$$

Por lo tanto si $b\neq 0$, el sistema es incompatible y si $b=0$, el sistema es compatible simplemente indeterminado. Solución del sistema:

```
(%i23) Eq12:  -2*x + y + z = 0$
       Eq22:  x  -2*y  + z = 0$
       Eq32:  x + y -2*z  = 0$
(%i26) linsolve([Eq12, Eq22, Eq32], [x,y,z]);
solve: dependent equations eliminated: (3)
```

$$(\%o26)\ [x=\%r8, y=\%r8, z=\%r8]$$

<u>Tercer caso</u>. Se sabe que el sistema es compatible determinado y entonces se puede calcular ya la solución:

```
(%i27) linsolve([Eq1, Eq2, Eq3], [x,y,z] );
```

$$(\%o27)\ [x=\frac{b}{a+2}, y=\frac{b}{a+2}, z=\frac{b}{a+2}]$$